

8.6 单总线(1-Wire)接口和温度计

“单总线”接口是美国 DALLAS 公司的特有技术(<http://www.dalsemi.com>),它实现了在一条数据线上进行双向数据传输,最大限度的节省了通讯线的数量。使系统布线更方便,成本更低,适合于多点分散系统。同时, DALLAS 公司推出了丰富的单总线产品,如数字温度传感器、电子纽扣(Ibutton)等。这种单总线芯片仅有一根信号线和一根地线。在信号线上综合了双向数字信号线和电源线的功能,每一块芯片都可提供一个独一无二的登记号,使用户可以灵活的构成不同功能的系统。而且,单总线产品基本为单芯片,成本低,可靠性高,例如: DS-1820,集传感器、温度转换、联网通讯于一体。

由于单总线仅使用一根信号线进行双向数据传输,因此单总线的通信协议就比其它的串行通信协议要复杂的多,通信时序和时间定时要求非常严格。在一般的电子产品开发中往往要花费大量的精力和时间编写和调试单总线接口的底层及通信程序。BASCOM-AVR 提供了简便的 1Wire 语句,可以用于方便和快速地开发单总线产品。

目的: 用 DS-1820 和 AVR 系统构成 0.1 度精度的温度计。

原理: 美国 Dallas 公司生产的单总线数字温度传感器 DS1820 可把温度信号直接转换成串行数字信号供微机处理。由于每片 DS1820 含有唯一的硅串行数,所以在一条总线上可挂接任意多个 DS1820。从 DS1820 读出的信息或写入 DS1820 的信息,仅需要一根数据线(单总线接口)。读写及温度变换功率来源于数据总线,总线本身也可以向所挂接的 DS1820 供电,而无需额外电源。DS1820 提供九位温度读数,构成多点温度检测系统而无需任何外围硬件。本实验采用一片 DS1820 与 AVR 单片微控制器和 LCD 显示器组成一个温度检测系统。

1. DS1820 的特性

单线接口: 仅需一根数据线与 MCU 连接无需外围元件可由总线提供电源;

测温范围为 $-55^{\circ}\text{C} \sim 75^{\circ}\text{C}$, 精度为达 0.1°C ;

内置 A/D 变换, 转换时间为 200ms; 九位温度读数,

用户自设定温度报警上下限, 其值是非易失性的;

2. DS1820 引脚及功能

DS1820 引脚见图 8.6.1(PR35 封装)。

GND: 地

DQ: 数据输入/输出脚(单总线接口, 可作寄生供电)

VDD: 电源电压。

3. DS1820 操作指令及时序

参见 Dallas 公司 DS1820 数据手册。




图 8.6.1




图 8.6.2

原理图：图 8.6.2

程序：Exp8.bas

\$regfile = "8515def.dat"

\$crystal = 4000000

Deflcdchar 0 , 14 , 10 , 14 , 32 , 32 , 32 , 32 , 32 ‘定义度’ 的显示字符

Deflcdchar 1 , 4 , 10 , 17 , 4 , 31 , 2 , 4 , 8 ‘定义汉字’ 今’ 的显示字符

Deflcdchar 2 , 31 , 4 , 31 , 4 , 12 , 10 , 18 , 17 ‘定义汉字’ 天’ 的显示字符

Dim I As Byte , Tmp As Byte , Crc As Byte

Dim T As Integer , T1 As Integer

Dim Data1(9) As Byte

Config 1wire = Portc.0 ‘Portc.0 接 DS1820 的数据线

Config

Lcdpin=Pin,Db4=Porta.4,Db5=Porta.5,Db6=Porta.6,Db7=Porta.7,E=Porta.0,Rs=Porta.1

Config Lcd = 16 * 2 ‘定义 LCD 接口

Cursor Off Noblink ‘清 LCD 显示，无光标

Cls

Locate 1 , 1 : Lcd "Demo for DS1820" ‘第一行显示提示字符串

Locate 2 , 1 : Lcd Chr(1) ; Chr(2) ‘第二行 1, 2 列显示汉字’ 今天’

Locate 2 , 8 : Lcd Chr(0) ; "C" ‘第二行 8 列显示字符’ ° C’

Do

 1wwrite &HCC : 1wwrite &H44 ‘写入 DS1820 命令，启动温度转换

 Waitms 255

 Waitms 255 ‘等待转换结束

 1wreset ‘DS1820 初始化

 1wwrite &HCC ‘写入 DS1820 命令，跳过 ROM 操作

 1wwrite &HBE ‘写入 DS1820 命令，读温度值

 Data1(1) = 1wread(9) ‘读 DS1820 温度，共 9 个字节

 1wreset ‘DS1820 初始化

```

If Err = 1 Then
  Locate 2 , 3 : Lcd "-----"          ‘无 DS1820 显示” -----”
Else
  Crc = 0
  For I = 1 To 9                          ‘进行 CRC 校验
    Tmp = Crc Xor Data1(i)
    Crc = Lookup(tmp , Crc8)
  Next I
  If Crc = 0 Then                          ‘CRC 校验正确，进行温度换算
    Tmp = Data1(1) And 1
    If Tmp = 1 Then Decr Data1(1)
    T = Makeint(data1(1) , Data1(2))
    T = T * 50 : T = T - 25
    T1 = Data1(8) - Data1(7) : T1 = T1 * 100
    T1 = T1 / Data1(8) : T = T + T1 : T = T / 10
  End If
  If Crc = 0 Then                          ‘CRC 校验正确，温度显示
    If T < 0 Then
      T = Abs(t)
      Locate 2 , 3 : Lcd "-"              ‘负温度显示” -” 号
    Else
      Locate 2 , 3 : Lcd " "              ‘正温度显示” “空格
    End If
    T1 = T Mod 10 : T = T / 10              ‘以下显示温度，精度 0.1 度
    If T < 10 Then
      Locate 2 , 4 : Lcd " " ; T ; "." ; T1
    Else
      If T > 99 Then
        Locate 2 , 3 : Lcd T ; "." ; T1
      Else
        Locate 2 , 4 : Lcd T ; "." ; T1
      End If
    End If
  End If
Else
  Locate 2 , 3 : Lcd "****.*"            ‘CRC 校验错，显示错误标志” ****.*”
End If
End If
Loop
End

Crc8:                                     ‘CRC 计算用的表格
Data 0 , 94 , 188 , 226 , 97 , 63 , 221 , 131 , 194 , 156
Data 126 , 32 , 163 , 253 , 31 , 65 , 157 , 195 , 33 , 127
Data 252 , 162 , 64 , 30 , 95 , 1 , 227 , 189 , 62 , 96
Data 130 , 220 , 35 , 125 , 159 , 193 , 66 , 28 , 254 , 160

```

Data 225 , 191 , 93 , 3 , 128 , 222 , 60 , 98 , 190 , 224
 Data 2 , 92 , 223 , 129 , 99 , 61 , 124 , 34 , 192 , 158
 Data 29 , 67 , 161 , 255 , 70 , 24 , 250 , 164 , 39 , 121
 Data 155 , 197 , 132 , 218 , 56 , 102 , 229 , 187 , 89 , 7
 Data 219 , 133 , 103 , 57 , 186 , 228 , 6 , 88 , 25 , 71
 Data 165 , 251 , 120 , 38 , 196 , 154 , 101 , 59 , 217 , 135
 Data 4 , 90 , 184 , 230 , 167 , 249 , 27 , 69 , 198 , 152
 Data 122 , 36 , 248 , 166 , 68 , 26 , 153 , 199 , 37 , 123
 Data 58 , 100 , 134 , 216 , 91 , 5 , 231 , 185 , 140 , 210
 Data 48 , 110 , 237 , 179 , 81 , 15 , 78 , 16 , 242 , 172
 Data 47 , 113 , 147 , 205 , 17 , 79 , 173 , 243 , 112 , 46
 Data 204 , 146 , 211 , 141 , 111 , 49 , 178 , 236 , 14 , 80
 Data 175 , 241 , 19 , 77 , 206 , 144 , 114 , 44 , 109 , 51
 Data 209 , 143 , 12 , 82 , 176 , 238 , 50 , 108 , 142 , 208
 Data 83 , 13 , 239 , 177 , 240 , 174 , 76 , 18 , 145 , 207
 Data 45 , 115 , 202 , 148 , 118 , 40 , 171 , 245 , 23 , 73
 Data 8 , 86 , 180 , 234 , 105 , 55 , 213 , 139 , 87 , 9
 Data 235 , 181 , 54 , 104 , 138 , 212 , 149 , 203 , 41 , 119
 Data 244 , 170 , 72 , 22 , 233 , 183 , 85 , 11 , 136 , 214
 Data 52 , 106 , 43 , 117 , 151 , 201 , 74 , 20 , 246 , 168
 Data 116 , 42 , 200 , 150 , 21 , 75 , 169 , 247 , 182 , 232
 Data 10 , 84 , 215 , 137 , 107 , 53

8.7 I2C 总线接口和简易 IC 卡读写器

I2C 总线是 PHILIPS 公司推出的串行总线，是各种总线中使用信号线较少，并且具有自动寻址，多主机时钟同步和仲裁等功能很强的总线。因此，许多接口芯片如：RAM，LCD 驱动，时钟，A/D，D/A 都采用 I2C 接口。IC 卡在当今社会的各个行业有着广泛的应用，而且大多数的 IC 卡的接口都采用 I2C 总线。

目的：用 AVR 系统构成简易 IC 卡读写器。

原理：

I2C 串行总线使用两根信号线：一根双向的数据线 SDA；另一根是时钟线 SCL。所有连接到 I2C 总线上的设备的串行数据都接到总线的 SDA 线，各设备的时钟线 SCL 接到总线的 SCL。I2C 数据总线传输时序如图 8.7.1，关于 I2C 的详细内容请参考有关的书籍和资料。





图 8.7.1

AVR 系列的单片微控制器没有专用的 I2C 总线硬件接口，因此需要用两根 I/O 线来模拟实现 I2C 总线的功能。这就要编写相应的 I2C 底层程序模拟 I2C 数据总线的传输时序，实现 I2C 总线启动、停止、读、写及应答的功能。BASCOM-AVR 提供了专用的 I2C 语句，利用它们就能非常方便地设计和实现 I2C 总线的读写。

实验中采用的 IC 卡为 ATMEL 公司的 AT24C01A/2/4/8/16，它是一种通用的读写存储卡。该类 IC 卡上的芯片就是采用 I2C 总线接口的串行 CMOS EEPROM。关于 AT24CXX 系列芯片的详细资料可访问 <http://www.atmel.com> 网站。

AVR 系列的单片微控制器内部还提供了一定容量的 EEPROM，可以用于保存系统的初始数据，设定值或密码口令字等，这一性能为开发产品提供了好多的方便，它不仅可使系统设计节省硬件 (EEPROM 芯片) 和连线；而且还减小了空间，提高了系统的可靠性和保密性。本设计中，使用了 AVR 片内的 EEPROM 来保存密码，并同 IC 卡上的密码核对，判别 IC 卡的非法性。

本设计将 IC 卡的读、写、判别结合在一起，用户使用 PC 的键盘输入 8 位自定密码，通过 RS-232 送到 IC 卡读写器，将密码写入用户的 IC 卡中 (也可同时写入 AVR 的 EEPROM 中作为系统密码)。IC 卡读写器还可读取 IC 卡上的密码，并同系统密码核对，用于判别 IC 的非法性。IC 卡读写器采用 LCD 液晶显示器，用于显示系统的状态和 IC 卡上的密码等。



原理图：见图 8.7.2




图 8.7.3 流程图

程序: Exp9.bas

```
$regfile = "8515def.dat"
```

```
$crystal = 4000000
```

```
$baud = 9600
```

```
Dim I As Byte , Temp As Byte
```

```

Dim Data1(8) As Byte , Data2(8) As Byte
Dim Card_ok As Bit

Config
Lcdpin=Pin,Db4=Porta.4,Db5=Porta.5,Db6=Porta.6,Db7=Porta.7,E=Porta.0,Rs=Porta.1
Config Lcd = 16 * 2          '定义 LCD 显示屏接口

Config Scl = Portd.7        '定义 Portd.7 为 I2C 总线的 Scl
Config Sda = Portd.6        '定义 Portd.6 为 I2C 总线的 Sda
Const Adresw = &HA0         '定义 IC 卡的写地址指令字
Const Adresr = &HA1         '定义 IC 卡的读地址指令字

Config Pinc.0 = Input , Pinc.1 = Input , Pinc.2 = Input
Portc.0 = 1 : Portc.1 = 1 : Portc.2 = 1      '定义 3 个 I/O 口为输入开关接口

Cursor Off Noblink        '清 LCD 显示

Do
  Cls
  Locate 1 , 1 : Lcd "Demo for IC_Card"      '显示提示字符
  If Pinc.2 = 1 Then                          '检测有无 IC 卡插入
    Locate 2 , 1 : Lcd "No IC_Card"         " '无 IC 卡插入
    Wait 1
  Else                                         '有 IC 卡插入
    If Pinc.0 = 0 Then                          '检测系统设置为写卡方式?
      Cls
      Locate 1 , 1 : Lcd "Enter Password: "   '写卡方式, 要求输入密码字
      I = 1
      Locate 2 , 1
      Do
        Temp = Inkey()                          '由 RS-232 口接收 PC 输入
        If Temp <> 0 Then                          '的密码字符,
          Data1(i) = Temp : I = I + 1             '长度为 8 个
          Lcd Chr(temp)                            '同时在 LCD 上显示密码
          Print Chr(temp);
        End If
      Loop Until I > 8
      Print
      For I = 1 To 8                               '将 8 个密码字写入 IC 卡中
        I2cstart                                  '写入地址为 1-8
        I2cwrite Adresw
        I2cwrite I
        I2cwrite Data1(i)
        I2cstop
    
```

```

        Waitms 10
    Next

    If Pinc.1 = 0 Then
        For I = 1 To 8
            Writeeprom Data1(i) , I
        Next
    End If

    For I = 1 To 8
        I2cstart
        I2cwbyte Adresw
        I2cwbyte I
        I2cstart
        I2cwbyte Adresr
        I2crbyte Data2(i) , Nack
        I2cstop
    Next

    Card_ok = 0
    For I = 1 To 8
        If Data1(i) <> Data2(i) Then
            Card_ok = 1
            Exit For
        End If
    Next

    Locate 2 , 1
    If Card_ok = 0 Then
        Lcd "Write Card ok! "
    Else
        Lcd "No or Bad Card!?"
    End If
    Do
        Loop Until Pinc.2 = 1
    Else
        Cls : Locate 1 , 1
    For I = 1 To 8
        I2cstart
        I2cwbyte Adresw
        I2cwbyte I
        I2cstart
        I2cwbyte Adresr
        I2crbyte Data1(i) , Nack

```



```
        I2cstop                                ‘读 IC 卡上的密码字 8 个
        Lcd Chr(data1(i))                      ‘显示在 LCD 上
    Next
    For I = 1 To 8
        Readeeprom Data2(i) , I                ‘读系统密码字 8 个
    Next

    Card_ok = 0
    For I = 1 To 8                              ‘将系统密码和 IC 卡上的密码
        If Data1(i) <> Data2(i) Then           ‘比较
            Card_ok = 1
            Exit For
        End If
    Next
    Locate 2 , 1
    If Card_ok = 1 Then                          ‘密码不符
        Lcd "Password not ok!"
    Else
        Lcd "Password Ok!      "              ‘密码相符
    End If
    Do
    Loop Until Pinc.2 = 1                        ‘等待用户将 IC 抽出
    End If
End If
Loop                                           ‘返回循环
End
```